# Геометрическая реконструкция как метод обучения. Геометрия пирамид Микерина и Хефрена


***Валентин Петрович Грибашев,***
*актёр, режиссёр*

***Татьяна Ивановна Кузнецова,***
*профессор Центра международного образования МГУ им. М.В. Ломоносова,*
*доктор педагогических наук, kuzti45@gmail.com*



*• геометрическая реконструкция памятников архитектуры • пирамида Микерина • пирамида Хеопса • пирамида Хефрена •*


Настоящая работа является продолжением статей[1], посвящённых геометрической реконструкции памятников архитектуры как методу обучения. В них геометрическая реконструкция предполагает не только восстановление первоначального облика археологического памятника, но и восстановление процесса его создания (предлагается вариант реконструкции разметочных работ древней египетской усыпальницы — пирамиды Джосера, и пирамиды Хеопса, самой знаменитой из трёх пирамид в Гизе). Продолжая и развивая разработанную там методику, предлагаем вариант реконструкции пирамид Микерина и Хефрена — двух других гизских пирамид.

## Основы реконструкции

Прежде чем приступать к этой реконструкции, напомним, что реконструкция пирамиды Хеопса основана на решении так называемого «вспомогательного» остроугольного треугольника (см. рис. 1б) со сторонами:

$$a = Rk,\ b = R / k,\ c = R\sqrt{2}, \qquad (1.1)$$

где $k = \sin 60°/\sin 75°$, $R = 10$ м — для «элементарного» вспомогательного треугольника, $R = 130$ м — для вспомогательного треугольника пирамиды Хеопса. Формулы (1.1) получены из следующих определений, сделанных нами на основе построений (см. рис. 1а) в квадрате со стороной $R$ при исследовании и реконструкции пирамиды Джосера:

$$a = |AC|,\ b = |AE|,\ c = |AD|. \qquad (1.2)$$




а б

*Рис. 1*

Реконструкция пирамиды Хеопса с помощью этого треугольника осуществляется следующим образом (см. рис. 2). Представим себе правильную четырёхугольную пирамиду, в основании которой — квадрат со стороной длины

[1] **Грибашев В.П., Кузнецова Т.И.** Геометрическая реконструкция памятников архитектуры как метод обучения // Школьные технологии. 2006. № 4. С. 85–90; **Грибашев В.П., Кузнецова Т.И.** Геометрия пирамиды Хеопса (геометрическая реконструкция как метод обучения) // Школьные технологии. 2008. № 4. С. 121–126.




а б

*Рис. 2*

2*a*, а величина угла наклона боковой грани к основанию равна β. Рассмотрим разрез этой пирамиды плоскостью, проходящей через высоту $SP_2$ пирамиды параллельно стороне основания.

Тогда каждый из двух вспомогательных треугольников $P_1P_4P_2$ и $P_3P_5P_2$ окажется внутри получившегося прямоугольного треугольника ($P_1SP_2$ и $P_3SP_2$ соответственно) со стороной длины *a*. Наша гипотеза заключается в том, что на рис. 2 изображён разрез пирамиды Хеопса — при *R* = 130 м. Чтобы убедиться в этом, необходимо сделать соответствующие расчёты и, прежде всего, «решить» вспомогательный треугольник рис. 1б.

Стороны вспомогательного треугольника заданы формулами (1.1), его углы вычисляются с помощью формул:

$\cos\alpha = (b^2 + c^2 - a^2) / 2bc$,

$$\alpha = \mathrm{arctg}\sqrt{1/\cos^2\alpha - 1}, \qquad (2)$$

$\cos\beta = (a^2 + c^2 - b^2) / 2ac$,

$$\beta = \mathrm{arctg}\sqrt{1/\cos^2\beta - 1}, \qquad (3)$$

$\cos\gamma = (a^2 + b^2 - c^2) / 2a$,

$$\gamma = \mathrm{arctg}\sqrt{1/\cos^2\gamma - 1}. \qquad (4)$$

Предлагается программа расчёта сторон и углов вспомогательного треугольника по этим формулам, написанная на алгоритмическом языке БЭЙСИК по методике пособия[2]:

**Программа 1**

```
10 REM РЕШЕНИЕ ВСПОМОГАТЕЛЬНОГО ТРЕУГОЛЬНИКА
20 PRINT «ВВЕДИТЕ R»
30 INPUT R
40 k=SIN(3.14159/3)/SIN(75*3.14159/180) : PRINT «k=»;k
50 a=R*k : b=R/k : c=R*SQR(2)
60 PRINT «a=»;a;«м », «b=»;b;«м », «c=»;c;«м »
70 x=b : y=c : z=a
80 PRINT «УГОЛ АЛЬФА=»; : GOSUB 140
90 x=a : y=c : z=b
100 PRINT «УГОЛ БЕТА =»; : GOSUB 140
110 x=a : y=b : z=c
120 PRINT «УГОЛ ГАММА =»; : GOSUB 140
130 GOTO 200
140 REM ПОДПРОГРАММА ВЫЧИСЛЕНИЯ УГЛА ТРЕУГОЛЬНИКА
150 KO=(x^2+y^2–z^2)/(2*x*y)
160 U=ATN(SQR(1/(KO*KO)–1)) : U0=U*180/3.14159
170 U1=INT(U0) : U2=(U0–U1)*60 : U3=INT(U2) : U4=(U2–U3)*60
180 PRINT U; «рад=»; U0;«град=»;U1;«град» ;U2;«мин=»;U1;«град»;U3;«мин»;U4;«сек»
190 RETURN
200 END
```

[2] **Брычков Е.Ю., Кузнецова Т.И.** Введение в информатику / Под общ. ред. Т.И. Кузнецовой. М.: УРСС, 1997. 208 с.

Здесь для большинства переменных приняты естественные обозначения, совпадающие с введёнными ранее, за исключением обозначений косинуса (KO) и значений величин углов (U, U0, U1, U2, U3, U4).

В строках 40–60 вычисляются длины сторон, далее — величины углов, при этом используется подпрограмма вычисления угла треугольника (строки 140–190). Так как программное обеспечение компьютера обеспечивает работу только с радианным выражением величин углов, то после вычисления величин углов мы воспользовались формулой перехода от радианной меры к градусной (см. строку 160). Так как при этом доли градуса получаются в десятичном виде, то при выводе результата мы были вынуждены использовать формулы перевода десятичного выражения долей градуса в минуты, а затем десятичных долей минуты в секунды (см. строку 170). В последнем случае два раза использовалась стандартная функция целой части (INT) при следующих обозначениях: U — величина угла в радианах; U0 — величина этого угла в градусах, представленная в десятичном виде; U1 — количество полных градусов; U2 — количество минут, содержащихся в дробной части градусной меры U0; U3 — количество полных минут в U2; U4 — количество секунд, содержащихся в дробной части минутной меры U2.

После ввода в компьютер значения $R$ = 10 м мы решили элементарный вспомогательный треугольник:

k = 0.8965753 (5)

a = 8.965753 м b = 11.15355 м
c = 14.14214 м (6)

УГОЛ АЛЬФА = .6864469рад = 39.33054град = 39град 19мин 49.95667сек (7)

УГОЛ БЕТА = .908284рад = 52.04088град = 52град 2мин 27.1756 сек (8)

УГОЛ ГАММА = 1.546862рад = 88.62872град = 88град 37мин 43.40332 сек (9)

Затем, введя в компьютер $R$ = 130 м, получили:

a = 116.5548 м b = 144.9962 м c = 183.8478 м

После округления с точностью до 0,5 мм пришли к окончательным результатам:

$a \approx 116{,}555$ м; $b \approx 144{,}996$ м; $c \approx 183{,}848$ м.

Ввиду необходимости для дальнейшего исследования мы вычислили длину $H_0$ высоты вспомогательного треугольника, опущенной на сторону с длиной $a$, и длину $H$ высоты $SP_2$ реконструкции рис. 2. При этом мы воспользовались формулами:

$$H_0 = c \cdot \sin\beta = R\sqrt{2}\,\sin\beta, \qquad (10.1)$$

$$H = a \cdot \mathrm{tg}\beta = R \cdot k \cdot \mathrm{tg}\beta. \qquad (10.2)$$

Для вычисления этих длин с помощью компьютера достаточно в программу 1 добавить строки:

```
101 PRINT «H0 =»; R*SQR(2)*SIN(U);«м»
102 PRINT «H=»;R*K*TAN(U);«м»
```

В результате для элементарного вспомогательного треугольника, т. е. при $R$ = 10 м, получили:

$$H_0 \approx 11{,}15035 \text{ м} \approx 11\text{м } 15 \text{ см } 0{,}4 \text{ мм}. \qquad (11.1)$$

А для пирамиды Хеопса, т. е. при $R$ = 130 м,

$$H \approx 149{,}4027 \text{ м} \approx 149 \text{ м } 40 \text{ см } 3 \text{ мм}. \qquad (11.2)$$

Итак, для пирамиды Хеопса мы получили:

1) длина стороны основания равна

$2a = 2 \cdot 116{,}555$ м = 233, 11 м;

2) длина высоты равна $H$ = 149,403 м;

3) величина угла наклона боковой грани к плоскости основания равна $\beta = 52°2'27''$.

При сравнении этих, полученных нами, значений параметров пирамиды Хеопса с известными в литературе, оказалось, что относительные погрешности не превышают 2 %. Таким образом, наша гипотеза подтвердилась.

Теперь можно перейти к реконструкции двух других знаменитых пирамид — Микерина и Хефрена.

## Пирамида Микерина

Возьмём $R$ = 60 м и введём его в программу 1. В результате компьютер выдаст $a$ = 53,795 м, что вполне соответствует половине длины стороны основания пирамиды Микерина. Действительно, известно, что длина стороны её основания может быть равна 108,04 м[3] или 108,5 м[4]. В нашем случае получается близкое значение — 107,59 м.

Чтобы построить вертикальный разрез, проходящий через высоту пирамиды параллельно стороне её основания (другими словами — через апофемы двух противоположных боковых граней пирамиды), как и при реконструкции пирамиды Хеопса, воспользуемся модулем 10 м и вспомогательным треугольником, рассчитанным с помощью компьютера.

Как и в случае пирамиды Хеопса, построим два вспомогательных треугольника (см. рис. 3), только за вершину пирамиды возьмём середину $S_0$ отрезка $P_4P_5$ (см. рис. 3а). Эту точку можно было получить и как пересечение высоты $SP_2$ с отрезком $P_4P_5$ (см. рис. 2) или без построения точки S, просто соединив точки $P_4$ и $P_5$ отрезком, а затем разделив его пополам. Далее, соединив точку $S_0$ с точками $P_1$ и $P_3$, получим разрез пирамиды Микерина (см. рис. 3б).

Вычислить важные для нас параметры этой пирамиды (длину высоты и величину угла наклона боковой грани к основанию) легко, посмотрев на рис. 3: искомая высота равна высоте $H_0$ вспомогательного треугольника, например, треугольника $P_1P_2P_4$, проведённой из вершины $P_4$ (см. формулу (10.1)), а угол $\beta_1$ наклона боковой грани пирамиды к плоскости её основания можно определить из прямоугольного треугольника $P_1S_0P_2$ по формуле, полученной из (10.2):

$$\beta_1 = \mathrm{arctg}(H_0/a). \qquad (12)$$

Составим программу расчёта пирамиды Микерина:

**Программа 2**

```
10 REM РАСЧЁТ ПИРАМИДЫ МИКЕРИНА
20 PRINT «ВВЕДИТЕ R»
30 INPUT R
40 k=SIN(3.14159/3)/SIN(75*3.14159/180)
50 a=R*k : b=R/k : c=R*SQR(2) : PRINT «a=»;a;«м »
60 KO=(a^2+c^2–b^2)/(2*a*c)
70 U=ATN(SQR(1/(KO*KO)–1))
80 H0=R*SQR(2)*SIN(U) : PRINT «H0=»;H0; «м »
```




а б

*Рис. 3*

[3] **Проскуряков С.Б.** Строители пирамид из созвездия Большого Пса. Орёл: Книга, 1992. С. 46.

[4] **Целлар К.** Архитектура страны фараонов: Жилище живых, усопших и богов / Пер. с венг. А.Д. Рагимбекова; Под ред. В.Л. Глазычева. М.: Стройиздат, 1990. С. 70. (Научно-популярная библиотека школьника).

```
90 V = ATN(H0/a) : V0=V*180/3.14159 : V1=INT(V0) : V2=(V0–V1)*60 : V3=INT(V2) : V4=(V2–V3)*60
100 PRINT «УГОЛ БЕТА 1=»; V0; «град=»; V1; «град»; V2; «мин=»; V1; «град»; V3; «мин»; V4; «сек»
110 END
```

Здесь, в основном, сохраняются обозначения программы 1, только U обозначает величину угла $\beta$, KO — $\cos\beta$. Естественно, H0 — это длина высоты пирамиды, V — величина угла $\beta_1$ в радианах, V0 — величина этого угла в градусах, представленная в десятичном виде, V1 — количество полных градусов, V2 — количество минут, содержащихся в дробной части градусной меры V0, V3 — количество полных минут в V2, V4 — количество секунд, содержащихся в дробной части минутной меры V2.

Запустив программу для $R$ = 60 м, получаем:

a=53.795 м

H0=66.90215 м

УГОЛ БЕТА 1 = 51.19806град = 51град11.88377мин = 51град11мин53.02643сек.

После разумного округления оказывается, что высота пирамиды Микерина равна 66,9 м, а величина угла наклона боковой грани к плоскости основания равна 51°12′. Эти результаты достаточно близки к известным из литературы: 66,4 м и 50°53′[5]; 66,5 м и 51°20′[6]. Вообще, относительная погрешность для стороны основания находится в диапазоне между 0,4 % до 0,84 %, для высоты — в диапазоне между 0,6 % и 0,75 %, для угла наклона боковой грани к плоскости основания — в диапазоне между 0,26 % и 0,62 %.

## Пирамида Хефрена

Реконструкция этой пирамиды использует элементы реконструкций, как пирамиды Хеопса, так и пирамиды Микерина.

Начнём, как и во всех других случаях, с основания. В литературе упоминаются следующие его размеры: 215,3 м × 215,3 м [4, с. 46]; 210,5 м × 210,5 м [5, с. 72]. Разделив первый размер пополам, получим 107,65 м. Сформулируем гипотезу о том, что если сделать расчёты с помощью программы 2, только не при $R$ = 60 м, а при $R$ = 120 м, то получим $a$ = 107,59 м, что очень близко к первому из упомянутых выше размеров.

Для высоты пирамиды Хефрена известен размер — 143,5 м. И здесь сформулируем гипотезу о том, что приблизительно такую высоту мы можем получить, если за неё возьмём высоту $S_0P_2$, построенную аналогично высоте пирамиды Микерина (см. рис. 3), которая равна высоте вспомогательного треугольника, но не при $R$ = 60 м, а при $R$ = 130 м.

Разрез пирамиды Хефрена по апофемам противоположных боковых граней — равнобедренный треугольник с вершиной $S_0$ и с основанием $2a$, где $a$ задаётся (по гипотезе) формулой (1.1) при $R$ = 120 м. Острый угол $\beta_2$ при основании этого треугольника — это угол наклона боковой грани пирамиды к её основанию. Чтобы вычислить его значение, воспользуемся методикой предыдущего пункта.

Для вычисления длин стороны основания, высоты пирамиды Хефрена и величины угла наклона боковой грани к основанию пирамиды, воспользуемся программой 3, составленной на основе предыдущих программ:

**Программа 3**

```
10 REM РАСЧЁТ ПИРАМИДЫ ХЕФРЕНА
20 PRINT «ВВЕДИТЕ RO, R»
30 INPUT RO, R
40 k=SIN(3.14159/3)/SIN(75*3.14159/180)
45 aO=RO*k : PRINT «aO=»;aO;«м »
50 a=R*k : b=R/k : c=R*SQR(2)
60 KO=(a^2+c^2–b^2)/(2*a*c)
70 U=ATN(SQR(1/(KO*KO)–1))
80 H0=R*SQR(2)*SIN(U) : PRINT «H0=»;H0; «м »
90 V = ATN(H0/aO) : V0=V*180/3.14159 : V1=INT(V0) : V2=(V0–V1)*60 : V3=INT(V2) : V4=(V2–V3)*60
```

100 PRINT «УГОЛ БЕТА 2=»; V0; «град=»; V1; «град»; V2; «мин=»; V1; «град»; V3; «мин»; V4; «сек»

110 END

Здесь появляются aO — длина половины основания пирамиды Хефрена и RO — соответствующее значение *R* (равное 120 м).

Итак, введя в компьютер значения RO = 120 м (для расчёта основания пирамиды) и R = 130 м (для расчёта высоты пирамиды), получаем результат:

aO = 107.589 м

H0 = 144.9547 м

УГОЛ БЕТА 2 = 53.41631град =
53град24.97833 мин =
53 град24мин58.69995сек.

После округления приходим к упомянутой в начале пункта длине стороны основания пирамиды Хефрена, равной 2 · 107,59 = 215, 18 м, её высоте 144, 955 м и к значению величины угла наклона боковой грани к основанию, равному 53°25′. Из литературы известны значения величины этого угла, равные 52°20′[5] и 53°7′[6], как видим, достаточно близкие к полученному нами.

Относительные погрешности для основных трёх параметров пирамиды Хефрена следующие: для стороны основания — 0,05 %, для высоты — 1 %, для угла наклона боковых граней к плоскости основания — 0,6 %. Таким образом, наши гипотезы подтвердились.

## Общие расчётные модули

На рис. 4 приведены разрезы по апофемам противоположных боковых граней пирамид Хеопса, Хефрена и Микерина, совмещённые по высоте и центру основания.

Из рисунка видно со всей очевидностью присутствие в реконструкции общего модуля: вспомогательного треугольника, а в построении основания — общего отрезка AC с длиной 8,966 м, которому кратна длина малой стороны пирамиды Джосера и длины



*Рис. 4*

[5] **Проскуряков С.Б.** Строители пирамид из созвездия Большого Пса. Орёл: Книга, 1992. С. 204.

[6] **Цыпкин А.Г.** Справочник по математике для средней школы / Под ред. С.А. Степанова. М.: Наука. Гл. ред. физ.-мат. лит., 1979. С. 70, 74.

сторон оснований всех пирамид в Гизе. А поскольку все вспомогательные элементы получены, исходя из отрезка в 10 м, то именно отрезок длины 10 м является модулем, общим для всех пирамид.

## Методический подход к «теоретическому» изучению пирамид в Гизе

Рассматривая настоящее исследование как методическую разработку, можем предложить учащимся целый ряд вопросов и заданий, в частности, задание нарисовать с помощью компьютера вспомогательный треугольник, а также воспроизвести на экране дисплея все «теоретически» проведённые нами построения разрезов пирамид Хеопса, Хефрена и Микерина. Эта работа с ещё большей очевидностью покажет учащимся, что в основе всех построений лежит вспомогательный треугольник. Построив его, далее, как из кирпичиков, из размноженных вспомогательных треугольников складываем разрез пирамиды Хеопса, а затем, проведя всего несколько отрезков, получаем и разрезы пирамид Хефрена и Микерина. Тем самым, учащиеся практически удостоверяются в том, что вспомогательный треугольник действительно является строительным модулем.

Покажем компьютерное построение вспомогательного треугольника (см. рис. 5). Заметим, что построение мы провели с использованием графического оператора SCREEN. Соответствующая программа выглядит следующим образом:




*Рис. 5*

**Программа 4**

```
10 REM ПОСТРОЕНИЕ ВСПОМОГАТЕЛЬНОГО ТРЕУГОЛЬНИКА
20 SCREEN 12
30 a=8.96575: H0=11.15035 : T=1.28182
40 X0=H0/T
50 LINE (0,H0∗10)–(a∗10,H0∗10)
60 LINE (0,H0∗10)–(X0∗10,0)
70 LINE (X0∗10,0)–(a∗10,H0∗10)
80 END
```

В программе a, H0, как и прежде, обозначают длины $a$ и $H_0$ соответственно, X0 — абсциссу $x_0$ вершины A. Последняя определяется по формуле строки 40, где T обозначает tgβ. Числовые значения $a$, $H_0$ (строка 30) были вычислены нами ранее[7]. То же самое касается значения tgβ[8]. Получить его и в рамках настоящей статьи, если в программу 1 вставить строку

```
103 PRINT «T=»; TAN(U)
```

В настоящей программе оператор LINE использован для построения сторон вспомогательного треугольника ABC: в строке 50 — стороны BC, в строке 60 — стороны BA, в строке 70 — стороны AC.

Интересным может быть решение задачи компьютерной реконструкции разреза пирамиды Хеопса — оно, естественно, отличается от предложенного нами, поскольку даёт в руки учащимся совершенно иные средства, определяемые математическим обеспечением компьютера и вытекающими из него возможностями. Это станет предельно понятно, если предложить учащимся построить рисунки данной работы с помощью координатного метода и программы 4.

[7] **Целлар К.** Архитектура страны фараонов: Жилище живых, усопших и богов / Пер. с венг. А.Д. Рагимбекова; Под ред. В.Л. Глазычева. М.: Стройиздат, 1990. С. 72. (Научно-популярная библиотека школьника).

[8] **Снисаренко А.** Гармония и алгебра Великой пирамиды // Техника — молодёжи. 1978. № 12. С. 204.




а б

***Рис. 6***

Вообще, обратим внимание на то, что построения, представленные на рисунках, осуществлялись с помощью компьютера в графическом редакторе. Для иллюстрации этого процесса из многих интересных ситуаций выделим вычерчивание биссектрисы угла COD на рис. 1а — путём «построения» ромба (см. рис. 6). Для этого (см. рис. 6.2) на луче OD строим отрезок OM, составленный из двух равных отрезков (половин $OM_0 = M_0M$). Чтобы на луче OC отметить точку $C_1$, такую, что $OC_1$=OM, дублируем одну половину отрезка OM, поворачиваем её на 90° и двигаем внутри угла COD до тех пор, пока её концы не окажутся на сторонах этого угла. Конец, находящийся на стороне OC, и будет искомой точкой $C_1$. Обозначим второй конец $M_1$. Затем, дублировав отрезок OM, передвинем его копию так, чтобы её левый конец попал в точку $C_1$. Тогда её правый конец $O_1$ будет лежать на биссектрисе угла COD. Далее остаётся только провести луч, являющийся биссектрисой, что делается без особого труда.

Чтобы обосновать проведённые построения, отметим, что:

1) угол COD равен 30°, так как треугольник AOB — равносторонний (см. рис. 6а). Поэтому на рис. 6б в прямоугольном треугольнике $OC_1M_1$ гипотенуза $OC_1$ равна двум катетам $C_1M_1$, т. е. отрезку OM. Таким образом, в четырёхугольнике $OC_1O_1M$ соседние стороны равны;
2) дублирование отрезка и передвижение его копии по плоскости рисунка означает их равенство и параллельность; поэтому
3) стороны OM и $C_1O_1$ равны и параллельны (по построению), следовательно, четырёхугольник $OC_1O_1M$ — параллелограмм.

Из 1) и 3) следует, что $OC_1O_1M$ — ромб и его диагональ $OO_1$ — биссектриса угла DOC, что и требовалось доказать.

Рассматривая настоящую работу как методическую разработку для проведения творческих работ с учащимися, необходимо отметить, что многое из подробно описанного в ней можно (по усмотрению преподавателя) подать учащимся в качестве как проблемных задач, так и задач на доказательство, задач на вычисление, оценочных задач, геометрических или чертёжных задач на построение (и их интегрированных вариантов), практических задач на алгоритмизацию разметочных процессов, задач на различные варианты получения результатов вычислений (ручной счёт плюс использование таблиц и с помощью компьютера). ❑